



AgEcon SEARCH
RESEARCH IN AGRICULTURAL & APPLIED ECONOMICS

The World's Largest Open Access Agricultural & Applied Economics Digital Library

This document is discoverable and free to researchers across the globe due to the work of AgEcon Search.

Help ensure our sustainability.

Give to AgEcon Search

AgEcon Search

<http://ageconsearch.umn.edu>

aesearch@umn.edu

*Papers downloaded from **AgEcon Search** may be used for non-commercial purposes and personal study only. No other use, including posting to another Internet site, is permitted without permission from the copyright owner (not AgEcon Search), or as allowed under the provisions of Fair Use, U.S. Copyright Act, Title 17 U.S.C.*



**MyGTAP Model: A Model for Employing Data from the MyGTAP
Data Application—Multiple Households, Split Factors,
Remittances, Foreign Aid and Transfers**

By

Terrie Walmsley

and

Peter Minor¹

GTAP Working Paper No. 78

2013

¹ Peter Minor is Managing Director of ImpactEcon, LLC and GTAP Research Fellow (email: peterminor@impactecon.com). Terrie Walmsley is a Research Associate Professor, Department of Agricultural Economics, Purdue University, USA and a Principal Fellow, Honorary Associate Professor in the Department of Economics, University of Melbourne, Australia (email: twalmsle@gmail.com).

Table of Contents

1. Introduction.....	4
2. The Removal of the Regional Household.....	5
3. Inter-regional Transfers.....	8
Remittances.....	9
Foreign Income.....	9
Foreign Aid.....	10
4. Multiple Households and Endowments	10
5. Private Household Expenditure.....	11
CDE	12
LES	13
6. Other Issues and Areas of Future Development	14
Armingtons	14
Population	15
Welfare	16
7. Appendixes.....	17
Appendix 1: List of Variables and Coefficients	17
Appendix 2: The Standard Closure and Some Alternative Closure Swaps	23
Appendix 3: A Note on Structural Singularity and Zeros in the Database	26
Appendix 4: LES Code.....	27
8. References.....	31

Important Acknowledgments and Conditions (must read)

MyGTAP was originally funded by the Bank of the Netherlands Partnership program, a joint program with the World Bank. William Martin, Team Leader and Director. This paper is based on a report submitted as part of the BNPP. Further improvements and features have been added. For the most up to date MyGTAP documentation, model, or programing code, visit: MyGTAP.ORG

The MyGTAP data programs are based on work by Mark Horridge of Monash University in his SplitCom programs found at:

<http://www.monash.edu.au/policy/splitcom.htm>

Some of the code in MyGTAP is extracted from SplitCom. The authors of MyGTAP wish to acknowledge Mark's contribution to this program.

MyGTAP Model is provided free of charge. By downloading and using the files the user is required to acknowledge their use in any publication or application by inserting into the code the following attribution: Walmsley, Terrie and Peter Minor "MyGTAP Model: A Model for Employing Data from the MyGTAP Data Application—Multiple Households, Split Factors, Remittances, Foreign Aid and Transfers", GTAP Working Paper Series, #78, Center for Global Trade Analysis, Purdue University, 2013. Analysis, reports, and publications in print or otherwise distributed utilizing the MyGTAP data program application and code, in whole or in part, will insert the aforementioned attribution in the first page of their publication after any cover page.

The user accepts full responsibility for the use of the code and agrees to hold the authors harmless for any errors or omissions. In accordance with the open software agreement: there is no warranty for the programs or documentation. The programs and documentation are provided "as is" without warranty of any kind, either expressed or implied, including, but not limited to, the implied warranties of merchantability and fitness for a particular purpose. The entire risk as to the quality and performance of the program is with you. Should the program prove defective, you assume the cost of all necessary servicing, repair or correction. In no event will the author/s who modifies and/or conveys the program be liable to you for damages, including any general, special, incidental or consequential damages arising out of the use or inability to use the program (including but not limited to loss of data or data being rendered inaccurate or losses sustained by you or third parties or a failure of the program to operate with any other programs).

1. Introduction

The purpose of this paper is to document an extension to the GTAP model (Hertel, 1997) to incorporate an alternative specification of the regional household as well as the addition of various inter-regional transfers². This new model was developed with the following aims: a) to give the user more flexibility in the treatment of government savings and spending; b) to include inter-regional transfers, such as remittances and foreign capital incomes; and c) to allow the user to model the impact of a policy on different households and factors within an economy/ies of interest for which they have additional data.

The following is a brief summary of the changes made in the development of this model:

1. The regional household has been removed and replaced with a separate government and private household:
 - a. Government now collects income from taxes and foreign aid and uses this money to consume (government expenditure). The difference between government income and consumption is the government deficit or saving.
 - b. Private households receive income from factors, including foreign remittances and capital and use this money to consume and save.
2. Inter-regional transfers such as labor remittances, foreign income on capital and foreign aid have been included into the model.
3. The model allows for multiple private households to be specified in one region or country³. Hence each private household holds endowments and receives income from those endowments.
4. Private household expenditure can now be allocated across commodities using either the constant difference elasticity (CDE) or linear expenditure system (LES) specifications. The method used can be altered by the user and can differ by country/region.

Each of these changes is discussed in turn below, followed by a section on areas for future development.

Note that this document is meant to be used in conjunction with a GTAP database splitting program (MyGTAP data program) which creates the database for use with this model⁴. We

² The model is written in GEMPACK, see Harrison and Pearson (2007)

³ Future extensions will allow multiple households in more than one region or country

⁴ MyGTAP is documented in Minor and Walmsley (2012).

strongly suggest that this document be reviewed in conjunction with the MyGTAP data program documentation before initial use.

2 The Removal of the Regional Household

In the GTAP model, a single regional household collects all income from factors and taxes and allocates this income to private consumption, government and (regional) savings using a Cobb Douglas function (see figure 1). Our first task was to remove this regional household and replace it with a separate government and private household (see figure 2).

The government is assumed to gain income from taxes (TTAX), foreign aid in (AIDI), less foreign aid out (AIDO) and other transfers (TRNG) from government to private households (equation 1)⁵. It is assumed that foreign aid is received directly by each country/region's government, not directly by private households.

$$GOVINC(r) = AIDI(r) - AIDO(r) + TTAX(r) - \text{sum}(h, HHL D, TRNG(h,r)) \quad (1)$$

Where $r \in \text{REG}$

$h \in \text{HHL D}$ (private households, assume this is a single private household)

This income (gincome in percentage changes) is then used by the government to fund government expenditure (yg, equation 2) and government savings (psave + qgsave or govdef, equation 3). The difference between government income and consumption is the government deficit or saving (govdef, equation 4). The user can assume that the share of government expenditure in income remains constant or they can specify alternative assumptions, such as fixing the government deficit⁶.

$$yg(r) - gincome(r) = dpgov(r) - dpgav(r) \quad (2)$$

$$psave(r) + qgsave(r) - gincome(r) = dpgsave(r) - dpgav(r) \quad (3)$$

$$govdef(r) = gincome(r) - yg(r) \quad (4)$$

⁵ Note this is a levels formula for manipulating the data. Upper case denotes levels, while lower case denotes percentage changes. A complete list of new variables and coefficients used in this paper are provided in Appendix 1. The tab files are also available on request.

⁶ By adding the following statement to the bottom of the closure page: Swap govdef = dpgsave ;

Figure 1: Standard GTAP Model





Figure 2: MyGTAP Model with Remittances, Foreign Aid and Transfer



Private households receive income from factors (EVOAH) less depreciation (VDEPH), plus net foreign labor remittances (REMIH and REMOH) and foreign capital income (FYIH and FYOH), transfers between households (TRNH) and from the government (TRNG) (Equation 5).

$$\begin{aligned}
 & \text{HHLDINC}(h,r) \\
 &= \text{sum}(i, \text{ENDW_COMM}, \text{EVOAH}(i,h,r)) - \text{VDEPH}(h,r) \\
 &\quad + \text{REMIH}(h,r) - \text{REMOH}(h,r) \\
 &\quad + \text{FYIH}(h,r) - \text{FYOH}(h,r) \\
 &\quad + \text{sum}(k, \text{HHLD}, \text{TRNH}(k,h,r) - \text{TRNH}(h,k,r)) \\
 &\quad + \text{TRNG}(h,r) \tag{5}
 \end{aligned}$$

Where $r \in \text{REG}$

$h \in \text{HHLD}$

Each private household's income is then allocated to private consumption and savings using Cobb Douglas as in the standard GTAP model.

Regional savings is calculated as the sum of private household savings and government savings (equation 6) and allocated across investment in the same way as the GTAP model. Note that the value of savings is no longer the same as that in the standard GTAP Database, because remittances and other foreign transfers have altered incomes, while expenditures on commodities remain the same. Savings is adjusted to ensure balance between income and expenditures.

$$\begin{aligned}
 & \text{SAVE}(r) * \text{qsave}(r) = \text{SAV_GOV}(r) * \text{qgsave}(r) \\
 & + \text{sum}(h, \text{HHLD}, \text{SAV_HHLD}(h,r) * \text{qhsave}(h,r)) \tag{6}
 \end{aligned}$$

3. **Inter-regional Transfers**

As discussed above a number of additional headers have been included in the GTAP Data Base to facilitate tracking of foreign ownership of factors and the incomes that flows from that ownership (labor/remittances and capital/foreign income); as well as foreign aid. This is based on work undertaken by Sonmez, McDonald and Walmsley (2011)⁷.

⁷ Further work will be undertaken over the next year to draw in more of the data discussed in this paper.

These foreign income flows are not bilateral, instead flows in and out of a country/region are provided. As a result assumptions need to be made regarding how these flows change as a result of a shock. Below we explain how this is done in the standard closure; shift variables are included so that these assumptions can be changed.⁸

Remittances

Remittances flowing out of a country (remoh) are assumed to change with average wages of skilled and unskilled labor (psh) and any changes in the endowment of labor (qoh) (Equation 7).⁹ The remittances in (remih) of every country then adjusts (by equation 8) to ensure that total remittances “in” equal total remittances “out” (Equations 9). Equation 9 determines remavi and all remittances in (remih) change by the average equation 8).¹⁰

$$\text{remoh}(h,r) = \text{sum}(i, \text{ENDWL_COMM}, \text{SHRLAB}(i,h,r) * [\text{psh}(i,h,r) + \text{qoh}(i,h,r)]) + \text{sremoh}(h,r) + \text{remavo} \quad (7)$$

$$\text{remih}(h,r) = \text{remavi} + \text{sremih}(h,r) \quad (8)$$

$$\text{sum}(h, \text{HHLD}, \text{sum}(r, \text{REG}, \text{REMI}(h,r) * \text{remih}(h,r))) = \text{sum}(h, \text{HHLD}, \text{sum}(r, \text{REG}, \text{REMO}(h,r) * \text{remoh}(h,r))) \quad (9)$$

By altering the closure remittances out can be fixed¹¹ or remittances in can be fixed and remittances out adjust to again ensure that total remittances in equal total remittances out.¹²

Foreign Income

Foreign income in and out is determined in the same way, although capital and rental rates are used to determine foreign income out rather than wages and labor supply (equations 10-12).

$$\text{fyoh}(h,r) = \text{sum}(i, \text{ENDWC_COMM}, \text{SHRCAP}(i,h,r) * [\text{psh}(i,h,r) + \text{qoh}(i,h,r)]) + \text{sfyoh}(h,r) + \text{fyavo} \quad (10)$$

$$\text{fyih}(h,r) = \text{fyavi} + \text{sfyih}(h,r) \quad (11)$$

$$\text{sum}(h, \text{HHLD}, \text{sum}(r, \text{REG}, \text{FYI}(h,r) * \text{fyih}(h,r))) = \text{sum}(h, \text{HHLD}, \text{sum}(r, \text{REG}, \text{FYO}(h,r) * \text{fyoh}(h,r)))$$

⁸ See Appendix 2 for the standard closure and some example alternative closures.

⁹ Variables sremoh and remavo are exogenous and equal to zero in the standard closure.

¹⁰ Variable sremih is exogenous and equal to zero in the standard closure.

¹¹ Swap sremoh = remoh;

¹² Swap remavi = remavo and Swap sremih = remih;

(12)

Foreign Aid

Foreign income in and out is determined in the same way, although government income is used to determine movements in foreign aid out (equations 13-15).

$$\text{aidout}(r) = \text{gincome}(r) + \text{saidout}(r) + \text{aidavo} \quad (13)$$

$$\text{aidin}(r) = \text{aidavi} + \text{saidin}(r) \quad (14)$$

$$\text{sum}(r, \text{REG}, \text{AIDI}(r) * \text{aidin}(r)) = \text{sum}(r, \text{REG}, \text{AIDO}(r) * \text{aidout}(r)) \quad (15)$$

4. Multiple Households and Endowments

As can be seen above, equations relating to private household income (equation 5), foreign income and remittances are allocated across households. The GTAP Database does not recognize multiple households, and hence there is only one private household (again, figure 1 illustrates the regional household in the standard GTAP model. All factor incomes accrue and all consumption and savings is undertaken by that one household, which we label the “Main Household” for the purposes of this model.¹³ Multiple households can be added to one country or region of the GTAP database using the MyGTAP data program.¹⁴

A number of changes are required to the model to include multiple households including: a) tracking of household factor supply and household ownership of factor endowments (income) and possible unemployment of those factors; b) the need to allow for additional endowment types; c) the need to accommodate transfers between households and to government; and d) the possibility of differential income and commodity taxes.

Each household now supplies endowments to firms. Hence the total supply of each endowment is the sum of the endowments supplied by all households (equation 16). This also

¹³ Note that this Main household is not the Regional Household discussed in Hertel (1997) because it is a private household. The GTAP regional household collects all income and allocates it to both private household and government consumption and savings. In contrast, Main household in MyGTAP simply aggregates all private households, excluding government transactions (see figure 2)

¹⁴ Note that this program produces a large number of zeros, for example when a household or endowment does not exist for a particular region. This can cause some structural singularity issues in the model. Appendix 3 contains a note explaining what was done to overcome these issues.

means that we know the ownership of capital by households (kbh) and hence can reduce household income by the appropriate amount of depreciation (equation 5).

$$qo(i,r) = \text{sum}(h, \text{HHL D}, \text{SHREVOMH}(i,h,r) * qoh(i,h,r)) \quad (16)$$

In order to incorporate unemployment closures we also include equations 17 and 18. $\text{emplh}(i,h,r)$ and $\text{empl}(i,r)$ allow for us to consider employment on labor supplied by particular households or all households equally.

$$qoh(i,h,r) = qoh_s(i,h,r) + \text{semp}lh(i,h,r) \quad (17)$$

$$\text{semp}lh(i,h,r) = \text{empl}h(i,h,r) + \text{empl}(i,r) \quad (18)$$

As in the standard GTAP model once the supply of each endowment ($qo(i,r)$) is determined this endowment moves easily or sluggishly between sectors depending on whether it is defined as a mobile or sluggish endowment.

The MyGTAP data program also allows the user to split endowments, since the existence of multiple households may also necessitate the need to separate, not just the supply, but also the demand for endowments. For example rural and urban households both supply unskilled workers, however, it is unlikely that these are easily substitutable and hence the user may want to separate the demand for rural unskilled and urban unskilled so as to reduce the substitutability between them (creating two prices which move independently in the processes). This is all achieved using the MyGTAP data program. Once the endowments are split in the data and the endowment set extended, the standard GTAP equations still hold and no changes are required to the underlying model.

The model also includes a number of additional variables to accommodate potential income transfer between households and differential taxes. Two transfers are included in the database and model, transfers between households $\text{TRNH}(k,h,r)$ is the transfer from household k to household h in region r ; and transfers from household h to the government ($\text{TRNG}(h,r)$). The user can specify the value of transfers in the MyGTAP data program, otherwise they are assumed to be zero. In the model these transfers are assumed to be exogenous. Income taxes (toh) and commodity taxes (tpdh and tpmh) are also included to allow for differences in tax rates paid by households.

5. Private Household Expenditure

The model is set up so that the user can define whether they want to use constant difference of elasticity (CDE) or linear expenditure system (LES) to determine household consumption of

each commodity¹⁵. This means that in developed economies where the Frisch parameter is one, the user can opt to keep GTAP's CDE, while imposing the LES in other regions where subsistence is important.

The user defines the use of CDE or LES through a binary parameter (PRIVTYPE) which is read in from the GTAPPARM file (default.prm). In the MyGTAP data program this takes the value of 1 for the special country where LES is to be used, and zero in all other countries¹⁶. The user can change this using ViewHAR but should be careful not to impose the LES on countries where the FRISCH parameter is greater than -1.8. The set of regions is then divided into two subsets:

Set REG_LES # Countries for which the LES system applies#

= (all,r,REG:PRIVTYPE(r)>0);

Set REG_CDE # Countries for which the CDE system applies#

= (all,r,REG: PRIVTYPE(r)=0);

First, regardless of the choice of specification, total private consumption expenditure ($y_{ph}(h,r)$) is determined by a Cobb Douglas function as private household income is allocated across private consumption and household savings in a similar way to which it is determined in the standard GTAP model, albeit at the household level (Equation 19).

$$y_{ph}(h,r) - p_{ph}(h,r) = [p_{privh}(h,r) + UELASPRIV(h,r) * u_{ph}(h,r)] \quad (19)^{17}$$

CDE

In countries where the CDE is used the traditional CDE equation in the standard GTAP model applies with two important differences (Equation 20). First, the equation only applies to the subset of regions REG_CDE; and second, household private expenditure ($y_{ph}(h,r)$) is being allocated across commodities not total private expenditure of the regional household ($y_p(r)$ in standard GTAP).

$$\begin{aligned} q_{ph}(i,h,r) - p_{ph}(h,r) = & \text{sum}(k,TRAD_COMM, EP(i,k,h,r) * p_{ph}(k,h,r)) \\ & + EY(i,h,r) * [y_{ph}(h,r) - p_{ph}(h,r)] \end{aligned} \quad (20)$$

¹⁵ See Hertel (1997) for an explanation of the CDE function used in GTAP.

¹⁶ This is similar to the parameter SLUG which is used for determining sluggish versus mobile endowments.

¹⁷ Note that UELASPRIV, the elasticity of private expenditure with respect to income is initially equal to 1 and is endogenous under both the LES and CDE specification. Under the CDE, this elasticity adjusts to reflect changes in the elasticity resulting from changes in income and the allocation of income across commodities with differing income elasticities. For more information on this see McDougall, XXXX. In the LES specification this elasticity remains unchanged.

where $r \in \text{REG_CDE}$

LES

The code used to incorporate LES for the REG_LES subset of countries is adapted from the Orani model developed by Dixon, Parmenter, Sutton and Vincent (1982). An extract of the code taken from the model is shown in Appendix 4.

First two parameters must be added to the MyGTAP model tab file:

1. The Frisch LES 'parameter' (FRISCH(h,r)) is calibrated from the income elasticity and household consumption shares¹⁸ or read in from the parameters file, if the header exists:

$$\text{FRISCH}(h,r) = \frac{\text{sum}\{c, \text{TRAD_COMM}, \text{CONSHR}(c,r) * \text{EY}(c,h,r) * [1.0 - \text{CONSHRH}(c,h,r) * \text{EY}(c,h,r)]\}}{[\text{EP}(c,c,h,r) + \text{CONSHRH}(c,h,r) * \text{EY}(c,h,r)]} \quad (21)$$

2. Household expenditure elasticities (EPS(i,h,r)) are set equal to the income elasticities also used in the CDE or read in from the parameters file, if the header exists.

$$\text{EPS}(c,h,r) = \text{EY}(c,h,r) \quad (22)$$

These parameters can then be used to determine the average (equation 23) and marginal (equation 24) share of luxury goods in total expenditure:

$$\text{SHR_LUX}(i,h,r) = \text{EPS}(i,h,r) / \text{ABS}[\text{FRISCH}(h,r)] \quad (23)$$

$$\text{BETA_LUX}(i,h,r) = \text{EPS}(i,h,r) * \text{CONSHRH}(i,h,r) \quad (24)$$

With the share of luxury goods in total expenditure known from equations (23) and (24) and total consumption expenditure determined by Equation (19) it is then a matter of determining how this income will be divided across subsistence ((qph_sub(i,h,r))) and luxury (qph_lux(i,h,r)) consumption. Total consumption (qph(i,h,r)) then depend on the sum of these two demands for subsistence and luxury commodities (equation 25).

$$\text{qph}(i,h,r) = \text{SHR_LUX}(i,h,r) * \text{qph_lux}(i,h,r) + [1 - \text{SHR_LUX}(i,h,r)] * \text{qph_sub}(i,h,r) \quad (25)$$

¹⁸ Calibration equations used are based on those taken from the CRUSOE suite developed by Mark Horridge. <http://www.monash.edu.au/policy/crusoe.htm> Note we also include an assertion that all FRISCH parameters are less than -1.8 for REG_LES countries.

Following the LES methodology, subsistence consumption ($qph_sub(i,h,r)$) remains constant and only changes with changes in the population or number of households ($poph(h,r)$) and any taste changes ($asub(i,h,r)$). This is shown in equation 26.

$$qph_sub(i,h,r) = poph(h,r) + asub(i,h,r) \quad (26)$$

Consumption of luxury commodities ($qph_lux(i,h,r)$) then depends on private expenditure left over for luxury consumption ($yph_lux(h,r)$), prices ($pph(i,h,r)$) and a taste parameter ($alux(i,h,r)$): Equation 27.

$$qph_lux(i,h,r) + pph(i,h,r) = yph_lux(h,r) + alux(i,h,r) \quad (27)$$

In order to determine how much of private expenditure is left for luxury goods (yph_lux) after the subsistence goods have been purchased we simply need to ensure that we are on our budget constraint (equation 28). That is, we need to ensure that total expenditure (yph , determined by equation 19) equals the sum of expenditures on all commodities, which depends on real consumption (qph , determined by equation 25) and prices (pph).

$$yph(h,r) = \text{sum}(i, \text{TRAD_COMM}, \text{CONSHRH}(i,h,r) * (qph(i,h,r) + pph(i,h,r))) \quad (28)$$

6. Other Issues and Areas of Future Development

A number of other changes were made to the model including allowing for region specific armingtons using the `ifheaderexists` command in Gempack; changes to population and finally the welfare decomposition was removed.

Armingtons

The implementation region-specific arminigtons in the `tab` file is shown in the box below.¹⁹ First `ESUBD_R`, the standard GTAP region-generic elasticity is defined and read into the model from the GTAP Database. Next a region-specific elasticity is defined. This is initially set equal to the region-generic, unless an additional header exists (`ifheaderexists`), “ESDR” containing region-specific details.

Coefficient (parameter)(all,i,TRAD_COMM)

¹⁹ The code is also adjusted in a similar way for the elasticity of substitution between imports from different regions.

ESUBD_R(i)
region-generic el. of sub. domestic/imported for all agents #;

Read

ESUBD_R from file GTAPPARM header "ESBD";

Coefficient (all,i,TRAD_COMM)(all,r,REG)

ESUBD(i,r) # region-specific el. of sub. among imports of i in Armington structure #;

Formula (all,i,TRAD_COMM)(all,r,REG)

ESUBD(i,r) = ESUBD_R(i) ;

Read (ifheaderexists)

ESUBD from file GTAPPARM header "ESDR";

In order to obtain region specific armingtons, the user can include them themselves directly in the parameters file or they can modify FlexAgg to aggregate the GTAP elasticities using region-specific weights. By using region-specific weights, the aggregated elasticities would differ across regions.²⁰

Population

The percentage change in the population by household ($poph(r)$) is included in the model as an exogenous variable. Since only the total population is known from the GTAP Database and not the populations of each household type, we cannot determine the percentage change in the total population from the changes by population; hence the percentage change in the total population ($pop(r)$) is removed from the model. This means that $ug(r)$ is no longer defined as a per capita variable and hence we re-label it $qgov(r)$ to show that it is now defined as real government expenditure. At this stage we do not have any equations related to migration of people between households.

²⁰ There are plans to include this in the GTAPAgg program.

Welfare

Since the regional household has been removed the current welfare decomposition needs to be revised.²¹ For the time being it has been removed.

²¹ This is on the list for future work.

7. Appendixes

Appendix 1: List of Variables and Coefficients

Coefficients

GOVINC(r)	REG	Income earned by government in r
AIDI(r)	REG	Value of foreign aid flowing into region r
AIDO(r)	REG	Value of foreign aid flowing out of region r
TTAX(r)	REG	Tax receipts in region r
HHLDINC(h,r)	HHLD,REG	Income earned by household h in r
EVOAH(i,h,r)	ENDW,HHLD,REG	Value of income on endowment i owned by household h in region r
VDEPH(h,r)	HHLD,REG	Value of capital depreciation on capital owned by household h in region r
REMIH(h,r)	HHLD,REG	Value of remittances in to household h in region r
REMOH(h,r)	HHLD,REG	Value of remittances out of household h in region r
FYIH(h,r)	HHLD,REG	Value of foreign income received on capital owned by household h in region r
FYOH(h,r)	HHLD,REG	Value of foreign income paid by household h in region r
SAVE(r)	REG	Total regional savings
SAV_GOV(r)	REG	Government saving in region r
SAV_HHLD(h,r)	HHLD,REG	Savings by household h in region r
SHRLAB(i,h,r)	LAB,HHLD,REG	Share of labor i in total labor in household h in region r

SHRCAP(i,h,r)	ENDWC,HHLD,REG	Share of capital i in total capital in household h in region r
SHREVOMH(i,h,r)	ENDW,HHLD,REG	Share of household h in value of endowment i in region r
ESUBD_R(c)	TRAD	Region-generic elasticity of substitution between domestic/imported goods for all agents
ESUBM_R(c)	TRAD	Region-generic elasticity of substitution among imports of i in armington structure
ESUBD(c,r)	TRAD,REG	Region-specific elasticity of substitution between domestic/imported goods for all agents
ESUBM(c,r)	TRAD,REG	Region-specific elasticity of substitution among imports of i in armington structure

New variables²²

Government			
gincome(r)	REG	Government income in region r	Endogenous
gincomeslack(r)	REG	Slack variable in the expression for government income	Exogenous
DGOVTBAL(r)	REG	Change in the government surplus/deficit in region r	Endogenous
qgsave(r)	REG	Real government saving/deficit in region r	Endogenous
dpgav(r)	REG	Average distribution parameter shift for government in region r	Endogenous

²² Unless otherwise stated these variables are percentage changes in.

dpgsave(r)	REG	Government saving distribution parameter in region r	Exogenous
govtrans(r)	REG	Transfers from government to household h	Exogenous
c_TRNG(h,r)	HHL D,REG	Value of transfers from government to household h (levels variable)	Endogenous
realTRNG(h,r)	HHL D,REG	Value of transfers from government to household h relative to government income (change)	Exogenous
Private Households			
hhldincome(h,r)	HHL D,REG	Income of household h in region r net of depreciation	Endogenous
hincomeslack(h,r)	HHL D,REG	Slack variable in the expression for regional income	Exogenous
qhsave(h,r)	HHL D,REG	Real Savings by Household h in region r	Endogenous
dphsave_all(r)	REG	Household saving distribution parameter applied to all households equally	Exogenous
psh(i,h,r)	ENDW,HHL D,REG	Supply price of endowment i in region r owned by household h	Endogenous
qoh(i,h,r)	ENDW,HHL D,REG	Supply of endowment i in region r owned by household h	Endogenous
hhldtrans(k,h,r)	HHL D,HHL D,REG	Transfer from household h1 to household h2 in region r	Exogenous
c_TRNH(h1,h2,r)	HHL D,HHL D,REG	Value of transfers from h1 to h2 in region r (levels, change)	Endogenous

realTRNH(h1,h2,r)	HHLD,HHL D,REG	Value of transfers from h1 to h2 in region r relative to household income (change)	Exogenous
Remittances			
remoh(h,r)	HHLD,REG	Remittances out of household h in region r	Endogenous
sremoh(h,r)	HHLD,REG	Shifter to change the rate of remittances out	Exogenous
remavo		Average rate of remittances out	Exogenous
remih(h,r)	HHLD,REG	Remittances into household h in region r	Endogenous
sremih(h,r)	HHLD,REG	Shifter to change the rate of remittances in	Exogenous
remavi		Average rate of remittances in	Endogenous
Foreign Income			
fyoh(h,r)	HHLD,REG	Foreign income out of household h in region r	Endogenous
sfyoh(h,r)	HHLD,REG	Shifter foreign income out household h in region r	Exogenous
fyavo		Average foreign income out household h in region r	Exogenous
fyih(h,r)	HHLD,REG	Foreign income in household h in region r	Endogenous
sfyih(h,r)	HHLD,REG	Shifter foreign income in household h in region r	Exogenous
fyavi		Average foreign income into household h in region r	Endogenous

Foreign Aid			
aidout(r)	REG	Foreign aid out of region r	Endogenous
saidout(r)	REG	Shifter on aid out of region r	Exogenous
aidavo		Average foreign aid out	Exogenous
aidin(r)	REG	Foreign aid in region r	Endogenous
saidin(r)	REG	Shifter on aid into region r	Exogenous
aidavi		Average foreign aid in	Endogenous
Endowment Supply			
qoh_s(i,h,r)	ENDW,HHL D,REG	Supply of endowment i from household h in region r (including unemployed)	Exogenous
semplh(i,h,r)	ENDW,HHL D,REG	Employment of endowment i from household h in region r	Endogenous
emplh(i,h,r)	ENDW,HHL D,REG	Shifter on employment of endowment i from household h in region r	Exogenous
empl(i,r)	ENDW,REG	Shifter on employment of endowment i in region r (equal across all households)	Exogenous
Population			
poph(h,r)	HHL D,REG	Population by household h in region r	Exogenous
Taxes			
tpdh(c,h,r)	TRAD,HHL D,REG	Commodity-, source-, and household-specific shift in tax on private consumption of domestic	Exogenous

		commodity c in region r by household h	
$tpmh(c,h,r)$	TRAD,HHL D,REG	Commodity-, source-, and household-specific shift in tax on private consumption of imported commodity c in region r by household h	Exogenous
$toh(i,h,r)$	ENDW,HHL D,REG	Income tax in region r on endowment i owned by household h	Exogenous

Appendix 2: The Standard Closure and Some Alternative Closure Swaps

Standard Closure

Exogenous	
poph	Population
qoh_s	Supply of endowments
Pfactwld	Numeraire
psaveslack profitslack incomeslack	Slack variables
endwslack cgdslack tradslack	
gincomeslack hincomeslack	
ams atm atf ats atd	Technological change variables
aosec aoreg avasec avareg	
afcom afsec afreg afecom afesec afereg	
aoall afall afeall	
dppriv dpgov dpgsave dphsave dphsave_all	Distribution parameters
to tp tm tms tx txs tpdh tpmh toh	Tax instruments
emplh empl	Employment variables
sremoh sremih remavo	Remittance shifters
sfyoh sfyih fyavo	Foreign income shifters
saidout saidin aidavo	Foreign aid shifters
realTRNG realTRNH ;	Transfers between households and government

Long run Closure

swap qoh_s(ENDWC_COMM, HHLD,REG) = expand(ENDWC_COMM, HHLD,REG);

Closures related to Remittances, Foreign Income and Foreign Aid etc.

Fix outward remittances, foreign income and foreign aid respectively

swap sremoh = remoh ;

swap sfyoh = fyoh ;

swap saidout = aidout ;

Increase remittances in by 10%

swap sremih = remih ;

swap remavo = remavi ;

shock remih = uniform 10;

Fixed Government Balance relative to Government Income

swap DGOVTBALR = dpgsave ;

Fixed Real Government Saving

swap qgsave = dpgsave ;

Shock Government and Household Transfers

Household Transfers (change in nominal):

Swap realTRNH("hapr", "hanp", "Tanzania") = c_TRNH("hapr", "hanp", "Tanzania") ;

shock c_TRNH = uniform 1;

Household Transfers (change in real)

shock realTRNH("hapr", "hanp", "Tanzania") = uniform 1;

Government Transfers (change in nominal):

swap realTRNG("hapr", "Tanzania") = c_TRNG("hapr", "Tanzania") ;

shock c_TRNG = uniform 1;

Government Transfers (change in real)

shock realTRNG("hapr","Tanzania") = uniform 1;

Fix the Trade Balance Relative to Income

Where savings of all households adjusts equally:

swap DTBALR("Tanzania") = dphsave_all("Tanzania");

Alternatively if just want one household to adjust:

swap DTBALR("Tanzania") = dphsave("MainHHLD", "Tanzania");

Where investment adjusts:

swap DTBALR("Tanzania") = cgdslack("Tanzania");

Unemployment

swap empl("UnSkLab",REG) = pfactreal("UnSkLab",REG) ;

swap empl("unsk_rural",REG) = pfactreal("unsk_rural",REG) ;

Appendix 3: A Note on Structural Singularity and Zeros in the Database

The MyGTAP Data program includes the multiple households and additional endowments for the countries of interest, while the main household and GTAP endowments for the other regions remain unchanged. The reason for this is that substantial data is required to split households and endowments for all countries. This then provides the usual GTAP results for all regions except those where additional data has been incorporated using MyGTAP. Unfortunately this results in a large number of zeros in the database, which will result in structural singularity if modifications are not made. In order to fix this problem of structural singularity the shares where numbers are zero are replaced by the average shares of the main household (Example 1: equation 19 and 20) or of the GTAP endowment to which it is mapped (Example 2: equation 21 and 22). The implications of this are that results are obtained for all households and endowments regardless of whether the data exists. These results are averages because of the use of the average shares. It is up to the user to understand their data and hence know which results should be analyzed and discussed.

Example 1

$$\text{CONSHRH}(i,h,r) = \text{VPAH}(i,h,r) / \text{PRIVEXPH}(h,r) \quad (19)$$

Unless $\text{PRIVEXPH}(h,r)=0$, then:

$$\text{CONSHRH}(i,h,r) = \text{VPA}(i,r) / \text{PRIVEXP}(r) \quad (20)$$

Example 2

$$\text{SHREVOMH}(i,h,r) = \text{EVOMH}(i,h,r) / \text{sum}(k, \text{HHLD}, \text{EVOMH}(i,k,r)) \quad (21)$$

Unless $\text{VOM}(i,r) = 0$, then

$$\begin{aligned} \text{SHREVOMH}(i,h,r) = & \text{sum}(k, \text{ENDW_COMM: MAPENDW}(k)=\text{MAPENDW}(i), \\ & \text{EVOMH}(k,h,r)) / \text{sum}(k, \text{HHLD}, \text{sum}(l, \text{ENDW_COMM: MAPENDW}(l)=\text{MAPENDW}(i), \\ & \text{EVOMH}(l,k,r))) \end{aligned} \quad (22)$$

Appendix 4: LES Code

Coefficient (all,h,HHL D)(all,r,REG)

FRISCH(h,r)

Frisch LES 'parameter' dependednt on EY, EP and CONSHR #;

Formula (initial)(all,h,HHL D)(all,r,REG)

FRISCH(h,r) = $\text{sum}\{c, \text{TRAD_COMM}, \text{CONSHRH}(c,h,r) * \text{EY}(c,h,r) * [1.0 - \text{CONSHRH}(c,h,r) * \text{EY}(c,h,r)] / [\text{EP}(c,c,h,r) + \text{CONSHRH}(c,h,r) * \text{EY}(c,h,r)]\}$;

Read (ifheaderexists) FRISCH from file GTAPDATA header "FRSH";

! ifheaderexists used if user wants to specify their own FRISCH parameters.!

Coefficient (all,i,TRAD_COMM)(all,h,HHL D)(all,r,REG)

EPS(i,h,r) *# Household expenditure elasticities - calibrated from EY #;*

Formula (initial)(all,c,TRAD_COMM)(all,h,HHL D)(all,r,REG)

EPS(c,h,r) = $\text{EY}(c,h,r)$;

Read (ifheaderexists) EPS from file GTAPDATA header "XPEL";

! ifheaderexists used if user wants to specify their own EPS parameters.!

Coefficient (all,r,REG)

CHKFRISCH(r) *#Check Frisch Parm of LES country is >-1.8#;*

Formula (all,r,REG)

CHKFRISCH(r) = $\text{maxs}(h, \text{HHL D}, \text{FRISCH}(h,r))$;

Coefficient (integer,parameter)(all,r,REG)

PRIVTYPE(r) *# Use CDE or LES in private consumption #;*

Formula (initial)(all,r,REG)

PRIVTYPE(r) = $0 + \text{if}(\text{CHKFRISCH}(r) < -1.8, 1)$;

Read (ifheaderexists)

PRIVTYPE from file GTAPPARM header "PREG";

! ifheaderexists used if user wants to specify which countries are to use LES !

Set

REG_LES *# Countries for which the LES system applies#*

= (all,r,REG: PRIVTYPE(r)>0);

Set

REG_CDE *# Countries for which the CDE system applies#*

= (all,r,REG: PRIVTYPE(r)=0);

Assertion (all,h,HHLD)(all,r,REG_LES)

FRISCH(h,r)<-1.8 ;

! Note that if user specifies countries as LES in PREG. But the FRISCH of these countries is > -1.8, then the model will stop.

FRISCH should be a large negative for LES countries. !

!<

2-2.b Alternate-Household Demand System (LES) for Developing Countries.

>!

Variable (all,h,HHLD)(all,r,REG_LES)

Utility_LES(h,r) # Utility per household #;

Variable (all,h,HHLD)(all,r,REG_LES)

yph_lux(h,r) # Household - supernumerary demands #;

Variable (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)

qph_lux(i,h,r) # Household - supernumerary demands of commodity i #;

Variable (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)

qph_sub(i,h,r) # Household - subsistence demands of commodity i #;

Variable (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)

alux(i,h,r) # Taste Change, household luxury demands#;

Variable (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)

asub(i,h,r) #Taste Change, household subsistence demands#;

Variable (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)

s_sub(i,h,r) #Taste change shifter for subsistence #;

Update (change)(all,h,HHLD)(all,r,REG_LES)

FRISCH(h,r) = FRISCH(h,r) * [yph(h,r) - yph_lux(h,r)]/100.0;

Update(change)(all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)

EPS(i,h,r) = EPS(i,h,r)*[qph_lux(i,h,r)-qph(i,h,r)
+ yph(h,r) - yph_lux(h,r)]/100.0;

Coefficient (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)

SHR_LUX(i,h,r) # (Luxury/total expenditure) #;
Formula (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)
 SHR_LUX(i,h,r) = EPS(i,h,r)/ABS[FRISCH(h,r)];

Coefficient (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)
 BETA_LUX(i,h,r) # Marginal household budget shares #;
Formula (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)
 BETA_LUX(i,h,r) = EPS(i,h,r) * CONSHRH(i,h,r);

Equation E_qlux
 # Luxury demand for commodities #
 (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)
 qph_lux(i,h,r) + pph(i,h,r) = yph_lux(h,r) + alux(i,h,r);

Equation E_qsub
 # Subsistence demand for commodities #
 (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)
 qph_sub(i,h,r) = poph(h,r) + asub(i,h,r);

Equation E_qph
 #Total Household demand form composite commodities#
 (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)
 qph(i,h,r)=SHR_LUX(i,h,r)*qph_lux(i,h,r)+[1-SHR_LUX(i,h,r)]*qph_sub(i,h,r);

Equation E_utility_LES
 # Change in utility disregarding taste change terms #
 (all,h,HHLD)(all,r,REG_LES)
 utility_LES(h,r) = **sum**{i,TRAD_COMM, BETA_LUX(i,h,r)*qph_lux(i,h,r)};

Equation E_alux
 # Default setting for luxury taste shifter #
 (all,i,TRAD_COMM) (all,h,HHLD)(all,r,REG_LES)
 alux(i,h,r) = asub(i,h,r) - **sum**{k,TRAD_COMM, BETA_LUX(k,h,r)*asub(k,h,r)};

Equation E_asub
 # Default setting for subsistence taste shifter #
 (all,i,TRAD_COMM)(all,h,HHLD)(all,r,REG_LES)
 asub(i,h,r) = s_sub(i,h,r) -
sum{k,TRAD_COMM, CONSHRH(k,h,r)*s_sub(k,h,r)};

Equation E_yph_lux

Finds yph_lux

(**all**,h,HHLD)(**all**,r,REG_LES)

yph(h,r) =

sum(i, TRAD_COMM, CONSHRH(i,h,r) * (qph(i,h,r) + pph(i,h,r))) ;

8. References

Hertel, T. (1997) *Global Trade Analysis: Modeling and Applications*, Cambridge: Cambridge University Press.

Harrison, J. and Ken. R. Pearson. 2007. *GEMPACK User Documentation Release 8.0*. Centre of Policy Studies and Impact Project Monash University, Melbourne, Australia. April.

Minor, P., and T. L. Walmsley, 2012. "MyGTAP Data Application: Program for Customizing and Extending the GTAP Database: Multiple Households, Split Factors, Remittances and Foreign Aid".

Sonmez, Y., S. McDonald and T. L. Walmsley, 2011. "Augmenting GTAP Database: Are the data on intra institutional transactions important?", Presented at the 14th Annual Conference on Global Economic Analysis, Venice, Italy

Dixon, Peter, Brian Parmenter, John Sutton and Dave Vincent (1982). *ORANI: A multisectoral model of the Australian Economy*, North-Holland.